# $Id: PKGBUILD 150958 2012-02-24 10:06:30Z pierre $
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-xdriinfo
pkgver=1.0.4
pkgrel=3
pkgdesc="Query configuration information of DRI drivers"
arch=('i686' 'x86_64')
url="http://xorg.freedesktop.org/"
license=('custom')
depends=('libx11' 'libgl')
makedepends=('xorg-util-macros' 'glproto' 'mesa')
groups=('xorg-apps' 'xorg')
source=("http://xorg.freedesktop.org/archive/individual/app/xdriinfo-${pkgver}.tar.bz2")
sha1sums=('723482b3fc627d7cc6124fec43dd1c21d102daaa')

build() {
  cd "${srcdir}/xdriinfo-${pkgver}"
  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/xdriinfo-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -m755 -d "${pkgdir}/usr/share/licenses/${pkgname}"
  install -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/"
}
